AMMRC TN 76-3

NETTING ANALYSIS FOR FILAMENT-WOUND PRESSURE VESSELS

Technical Note by

DAVID K. ROYLANCE

August 1976

D/A Project 1W362306AH72
AMCMS Code 632306.11.H72
Hardened BMD Materials

COMPOSITES DIVISION
ARMY MATERIALS AND MECHANICS RESEARCH CENTER
Watertown, Massachusetts 02172

AMMRC Ex. O. 35905

INTRODUCTION

Filament winding is an efficient and often used means of fabricating pressure vessels, partly because it permits the designer to optimize the structure by placing only enough fibers along given directions to withstand the stresses along those directions. If homogeneous materials are used to fabricate closed-end pressure vessels, the longitudinal stresses are only half the hoop stresses, so that material is effectively wasted in the longitudinal direction. Because of the efficiencies offered by filament winding, this technique has been used for a wide variety of pressure vessel applications in the years since World War II; these include motor cases for nearly all U.S. strategic missile systems, and recently they have been considered for light man-portable antitank missile launchers.

The increase in efficiency offered by filament winding is accompanied by an increase in the complexity of the techniques which must be employed in vessel design and analysis of vessel performance. The elementary equations of strength of materials are not valid, since the assumptions of homogeneity and isotropy used in their derivation are not met. There is also a greater number of material variables which must be considered, e.g., the number of helical and hoop layers, and the angle of helical winding. The purpose of this note is to present an analytical technique applicable to filament-wound pressure vessels which can predict the internal stresses arising from a given combination of internal pressure and choice of materials and processing parameters.

Netting analysis[1,2] is perhaps the simplest of the analytical techniques used in rationalizing the behavior of fiber-reinforced composite materials, but it is well suited for filament-wound pressure vessels. Netting analyses assume that all loads are supported by the fibers only, neglecting any contribution by the matrix and any interaction between the fibers. These assumptions do not accrue in general — they predict, for instance, zero transverse strength in a unidirectionally reinforced plate — but they are consistent with pressure vessels, in which the fibers are loaded in tension and the shearing stresses are small in comparison.

DEVELOPMENT OF GOVERNING EQUATIONS

Closed-End Bottles

Closed-end refers to pressure vessels constructed so that internal pressure sets up both hoop and longitudinal stresses, as opposed to open-tube configurations in which the longitudinal loads are taken by the next fixture.

Consider a cylindrical section of a chamber of radius R containing an internal pressure P, with ϕ denoting the longitudinal (axial) and θ the circumferential (hoop) directions. The strain ε_α at an angle α from the longitudinal is given in terms of the axial hoop strains $\varepsilon_\phi, \varepsilon_\theta$ by the transformation equation:[3]

1. GARG, S. K., et al. *Analysis of Structural Composite Materials.* Marcel Dekker, Inc., New York, 1973, p. 11-19.
2. *Trident I C4 Program Monthly Status Report-January 1973.* Report SA005-B2AOOEDP-12, B-74, p. 3. Request through U.S. Navy. Strategic Systems Project Office.
3. HIGDON, A., et al. *Mechanics of Materials.* John Wiley & Sons, Inc., New York, 1967, p. 52.

$$\varepsilon_\alpha = \varepsilon_\phi \cos^2\alpha + \varepsilon_\theta \sin^2\alpha + \gamma_{\phi\theta} \sin\theta \cos\theta \quad (1)$$

where the term involving the shear strain $\gamma_{\phi\theta}$ will be neglected.

This development will assume the presence of two sets of helical windings oriented at $\pm\alpha_1$ and $\pm\alpha_2$, and one set of hoop windings. The stresses in these fibers are given directly from Equation 1 as

$$\sigma_h = E\varepsilon_\theta \quad (2)$$

$$\sigma_{\alpha 1} = E(\varepsilon_\theta \sin^2\alpha_1 + \varepsilon_\phi \cos^2\alpha_1) \quad (3)$$

$$\sigma_{\alpha 2} = E(\varepsilon_\theta \sin^2\alpha_2 + \varepsilon_\phi \cos^2\alpha_2) \quad (4)$$

where σ denotes stress and E is the fiber modulus.

Denote the total cross-sectional area of fibers in a unit length of hoop or the two helical windings as A_h, $A_{\alpha 1}$, and $A_{\alpha 2}$. (These areas can be computed in several ways for actual vessels; an example will be given later.) Then the area of helical fibers intersecting a unit length of surface along the hoop and axial directions are $A_\alpha \sin\alpha$ and $A_\alpha \cos\alpha$. The resolved components of force from the helical fibers in the hoop and axial directions are then

$$\text{hoop} = (A_\alpha \sin\alpha)\sigma_\alpha \sin\alpha$$

$$\text{axial} = (A_\alpha \cos\alpha)\sigma_\alpha \cos\alpha.$$

It now remains to establish force balances in the hoop and axial directions. First the hoop:



$$2RP = 2[A_h\sigma_h + (A_{\alpha 1} \sin\alpha_1)\sigma_{\alpha 1} \sin\alpha_1 + (A_{\alpha 2} \sin\alpha_2)\sigma_{\alpha 2} \sin\alpha_2]. \quad (5)$$

Now the longitudinal:



$$(\pi R^2)P = (2\pi R)[(A_{\alpha 1} \cos\alpha_1)\sigma_{\alpha 1} \cos\alpha_1 + (A_{\alpha 2} \cos\alpha_2)\sigma_{\alpha 2} \cos\alpha_2)]. \quad (6)$$

The stresses may now be written in terms of strains by means of Equations 2 to 4; Equations 5 and 6 then become:

$$PR = A_h E\varepsilon_\theta + A_{\alpha 1} \sin^2\alpha_1 E(\varepsilon_\theta \sin^2\alpha_1 + \varepsilon_\phi \cos^2\alpha_1) + A_{\alpha 2} \sin^2\alpha_2 E(\varepsilon_\theta \sin^2\alpha_2 + \varepsilon_\phi \cos^2\alpha_2) \quad (7)$$

$$PR/2 = A_{\alpha 1} \cos^2\alpha_1 E(\varepsilon_\theta \sin^2\alpha_1 + \varepsilon_\phi \cos^2\alpha_1) + A_{\alpha 2} \cos^2\alpha_2 E(\varepsilon_\theta \sin^2\alpha_2 + \varepsilon_\phi \cos^2\alpha_2) \quad (8)$$

which can be written as:

$$PR/E = K_3\varepsilon_\theta + K_1\varepsilon_\phi \quad (9)$$

$$PR/2E = K_1\varepsilon_\theta + K_2\varepsilon_\phi \quad (10)$$

where:

$$K_1 = A_{\alpha 1} \sin^2\alpha_1 \cos^2\alpha_1 + A_{\alpha 2} \sin^2\alpha_2 \cos^2\alpha_2 \quad (11)$$

$$K_2 = A_{\alpha 1} \cos^4\alpha_1 + A_{\alpha 2} \cos^4\alpha_2 \quad (12)$$

$$K_3 = A_h + A_{\alpha 1} \sin^4\alpha_1 + A_{\alpha 2} \sin^4\alpha_2. \quad (13)$$

The K's constitute a geometrical description of the vessel. Equations 9 and 10 are solved easily to give:

$$\varepsilon_\theta = PR/2E[(K_1 - 2K_2)/(K_1^2 - K_1K_3)] \quad (14)$$

$$\varepsilon_\phi = PR/2E[(2K_1 - K_3)/(K_1^2 - K_2K_3)]. \quad (15)$$

The magnitudes of the fiber stresses are then given directly by using the results of Equations 14 and 15 in Equations 2 to 4.

Open-End Tubes

For the case in which the longitudinal loads are reacted by the test fixture rather than the test specimen, the left hand side of Equation 6 is set to zero. Equations 9 and 10 then become

$$RP/E = K_3\varepsilon_\theta + K_1\varepsilon_\phi \quad (16)$$

$$0 = K_1\varepsilon_\theta + K_2\varepsilon_\phi \quad (17)$$

which yields

$$\varepsilon_\theta = PR/E[(K_2)/(K_2K_3 - K_1^2)] \quad (18)$$

$$\varepsilon_\phi = -K_1/K_2\varepsilon_\theta = -PR/E[(K_1)/(K_2K_3 - K_1^2)]. \quad (19)$$

As before, the fiber stresses can be obtained by using these values in Equations 2 to 4.

A BASIC-LANGUAGE PROGRAM FOR CLOSED-END CHAMBERS

General

In order to facilitate rapid and error-free implementation of the foregoing analysis, a small BASIC-language program has been implemented for the Hewlett-Packard 9830 calculator. This program outputs values of hoop and axial strains, and hoop and helical fiber stresses, given values of layer areas, helix angle, chamber radius, fiber modulus, and internal pressure. A program listing is presented in the Appendix. Conversion of the program to other computational systems would not be difficult.

Definition of Program Variables

Input Variables

T$	Title (Alphanumeric, but avoid quotation marks) 80 characters maximum
Z1, Z2	Helical winding angles, degrees
R	Chamber radius, inches
A1, A2	Helical layer areas, square inches
A3	Hoop layer area, square inches
E	Fiber modulus, mpsi
P	Internal pressure, psig

Internal Variables

K1, K2, K3 As defined in Equations 11 to 13

Output Variables

E1	Hoop strain, percent
E2	Axial strain, percent
S1	Fiber stress in #1 helical layer, ksi
S2	Fiber stress in #2 helical layer, ksi
S3	Fiber stress in hoop layers, ksi

User Instructions

1. Insert program cassette, rewind to start
2. Press LOAD, EXECUTE (program is at File #0)
3. Rewind cassette
4. Press RUN, EXECUTE
5. Enter input parameters as requested by calculator
6. After each problem is solved the user has the option of inputting a new pressure while leaving the chamber geometry unchanged, or starting a totally new problem.

EXAMPLE PROBLEMS

Analysis of Burst Test Data

As part of an Organic Materials Laboratory program investigating filament-wound Kevlar/epoxy as a possible material for interceptor missile motor cases, a series of 6-inch-diameter, 14-inch-long bottles have been fabricated using an XOXOO winding pattern (X denotes helical windings, O denotes hoop windings) with ±25° helix angle. These bottles are gaged to permit measurement of hoop and axial strains, then tested by hydrostatic pressurization. Once the burst pressure is known experimentally, the netting analysis can be used to calculate the fiber stress at failure. This delivered fiber strength is an important parameter in chamber design.

Before the netting analysis can be performed, it is necessary to calculate the cross-sectional fiber areas A_h, $A_{\alpha 1}$, and $A_{\alpha 2}$ of the hoop and helical layers. Accurate values of these areas could be obtained by quantitative microscopy of suitably sectioned chambers, but acceptable values can be calculated more conveniently from values of filament denier and winding bandwidth as follows:

The denier (grams weight per 9000 meters length) of the 12-end roving used in winding is 4560, or 5.067×10^{-3} g cm^{-1}. Kevlar's density is 1.45 g cm^{-3}, so the area of the roving is $(5.067 \times 10^{-3}/1.45) = 3.49 \times 10^{-4}$ cm^2 $= 5.42 \times 10^{-4}$ in.^{2}. The width of a single band is measured during winding as 0.135 to 0.145 in.; taking the bandwidth as 0.140 in., there are $(5.42 \times 10^{-4}/0.140) = 3.87 \times 10^{-3}$ in.^{2} of fiber area in each inch of pass length. In the XOXOO winding pattern, there are two helical and three hoop layers, each consisting of two winding passes. The total areas are then:

$$A_h = (3)\ (2)\ (3.87 \times 10^{-3}) = 0.0232 \text{ in.}^2/\text{in.}$$

$$A_{\alpha 1} = (2)\ (2)\ (3.87 \times 10^{-3}) = 0.0155 \text{ in.}^2/\text{in.}$$

$$A_{\alpha 2} = 0.$$

A general formula, using mixed but commonly used units is:

$$A = (1.722 \times 10^{-7})\ (\text{DEN})\ N/\rho\ B$$

where A is the resulting area in square inches, DEN is the roving denier, N is the number of passes, ρ is the fiber density in grams per cubic centimeter, and B is the bandwidth in inches.

Test bottle 4 experienced catastrophic failure at 3042 psig internal pressure. Output from the netting analysis program for this pressure and the above layer areas is shown in Figure 1. These predictions can be compared with the strain gage readings for axial and hoop strains, and the fiber stresses calculated from the gage readings and Equations 2 to 4. These values are:

	From Gages	Netting Analysis	Difference (%)
Hoop Strain, Percent	1.25	1.84	32
Axial Strain, Percent	0.69	1.89	63
Hoop Stress, ksi	238	351	62
Helical Stress, ksi	160	358	58

Agreement is poor. Netting analysis is generally expected to yield predictions approximately 10% higher than those of more sophisticated finite-element analyses,[2] but the large disagreements above are ascribed primarily to gaging

```
              PRESSURE VESSEL FIBER STRESS NETTING ANALYSIS

                        DAVE ROYLANCE JULY 75

KEVLAR 6 IN BOTTLE, XOXUO PATTERN, 25 DEG HELIX

HELIX ANGLES: ALPHA1= 25.0     ALFHA2= 0.0
VESSEL RADIUS= 3.0    IN
LAYER AREAS (SQ IN): A1= 0.015500          A2= 0.000000    A3= 0.023200
FIBER MODULUS= 19.0      MPSI

                  PRESSURE= 3042       PSIG
              % HOOP STRAIN= 1.845
             % AXIAL STRAIN= 1.895
      STRESS IN #1 HELICALS= 358.4     KSI
            STRESS IN HOOPS= 350.6     KSI
```

Figure 1. Program output for six-inch Kevlar bottle at 3042 psig.

errors. Reasons for suspecting the gage readings are as follows: (a) The XOXOO pattern was designed by Hercules, Inc., to be balanced, i.e., have equal hoop and helical fiber stresses. This equality obtains in the netting analysis, but not in the gage values. (b) A delivered fiber strength of approximately 350 ksi was measured in an intensive Hercules test program using identical bottles. The stresses implied by the gage readings appear much too low. (Note that the 350 ksi strength is much less than either the mean (525 ksi) or the minus 3σ (447 ksi) strength measured in impregnated Kevlar strand tests. This loss in delivered fiber strength is likely due to still unexplained failure mechanisms in the filament-wound bottles, and is a serious problem in motor case design.)

Parametric Design Studies

Although the netting analysis is valuable in analyzing burst test results and checking the validity of gage readings, it is perhaps even more useful as a means of designing pressure vessels. In this capacity the program is run repeatedly using various values of layer areas and helix angles on a trial-and-error basis until the desired fiber stresses are obtained. To reduce the number of trials it may be convenient to construct plots showing the influence of various parameters; the desired values may then be located easily.

As an example, Figures 2 and 3 show the influence of internal pressure and helix angle on the fiber stresses of 6-inch XOXOO Kevlar bottles having layer areas $A_h = 0.0216$, $A_{\alpha 1} = 0.0144$, and $A_{\alpha 2} = 0$. The helix angle in Figure 2 is ±25°. Figure 1 demonstrates the linear relation between pressure and fiber stress, while Figure 3 illustrates that a helix angle near 25° yields a balanced design (equal helical and hoop stresses).



Figure 2. Influence of internal pressure on helical and hoop fiber stresses for six-inch Kevlar bottles.



Figure 3. Influence of helix angle on helical and hoop fiber stresses for six-inch Kevlar bottles.

APPENDIX. PROGRAM LISTING

```
10 REM*********************************************************************
20 REM NETTING ANALYSIS - PRESSURE VESSEL FIBER STRESS
30 REM ROYLANCE JULY 75 BMD PROGRAM X.O. 35905
40 REM*********************************************************************
50 DIM T$[80]
60 PRINT LIN(2)
70 PRINT TAB17;"PRESSURE VESSEL FIBER STRESS NETTING ANALYSIS"
80 PRINT LIN(1)
90 PRINT TAB27;"DAVE ROYLANCE JULY 75"
100 PRINT LIN(1)
110 REM*********************************************************************
120 REM ENTER PROBLEM PARAMETERS
130 REM*********************************************************************
140 DISP "ENTER TITLE";
150 INPUT T$
160 PRINT T$,LIN(1)
170 DEG
180 DISP "ENTER HELIX ANGLES (ALPHA1,ALPHA2)";
190 INPUT Z1,Z2
200 FIXED 1
210 PRINT "HELIX ANGLES: ALPHA1=";Z1,"ALPHA2=";Z2
220 DISP "ENTER VESSEL RADIUS, IN";
230 INPUT R
240 PRINT "VESSEL RADIUS=";R;" IN"
250 DISP "ENTER LAYER AREAS, SQ IN (A1,A2,A3)";
260 INPUT A1,A2,A3
270 FIXED 6
280 PRINT "LAYER AREAS (SQ IN): A1=";A1,"A2=";A2,"A3=";A3
290 K1=A1*(SINZ1)↑2*(COSZ1)↑2+A2*(SINZ2)↑2*(COSZ2)↑2
300 K2=A1*(COSZ1)↑4+A2*(COSZ2)↑4
310 K3=A3+A1*(SINZ1)↑4+A2*(SINZ2)↑4
320 DISP "ENTER FIBER MODULUS, MPSI";
330 INPUT E
340 FIXED 1
350 PRINT "FIBER MODULUS=";E;" MPSI",LIN(1)
360 E=E*1E+06
370 DISP "ENTER PRESSURE, PSIG";
380 INPUT P
390 FIXED 0
400 PRINT TAB23;"PRESSURE=";P;" PSIG"
410 REM*********************************************************************
420 REM CALCULATE STRAINS AND STRESSES
430 REM*********************************************************************
440 E1=((P*R)/(2*E))*((K1-2*K2)/(K1↑2-K2*K3))
450 E2=((P*R)/(2*E))*((2*K1-K3)/(K1↑2-K2*K3))
460 FIXED 3
470 PRINT TAB17;" % HOOP STRAIN=";E1*100
480 PRINT TAB17;"% AXIAL STRAIN=";E2*100
490 S1=E*(E1*(SINZ1)↑2+E2*(COSZ1)↑2)
500 S2=E*(E1*(SINZ2)↑2+E2*(COSZ2)↑2)
510 S3=E*E1
520 FIXED 1
530 PRINT TAB10;"STRESS IN #1 HELICALS=";S1/1000;" KSI"
540 IF A2=0 THEN 560
550 PRINT TAB10;"STRESS IN #2 HELICALS=";S2/1000;" KSI"
560 PRINT TAB10;"       STRESS IN HOOPS=";S3/1000;" KSI",LIN(1)
570 REM*********************************************************************
580 REM BRANCH OPTIONS
590 REM*********************************************************************
600 DISP "ENTER 1 FOR NEW PRESSURE, 2 FOR NEW PROBLEM";
610 INPUT Q
620 GOTO Q OF 370,100
630 STOP
640 END
```